
I ) RÉSULTATS MATHÉMATIQUES GÉNÉRAUX
          (on suppose implicitement que toute intégrale citée est convergente)
Définitions
On nomme « polynôme trigonométrique de degré n » toute expression de la forme :

 ∑
k=−n

n

c k ei kθ   ou  ∑
k=0

n

{ak cos (nθ )+bk sin (nθ )}    

        

On nomme « série trigonométrique » toute expression de la forme :

 ∑
k=−∞

∞

ck ei kθ   ou  ∑
k=0

∞

{ak cos(nθ )+bk sin (nθ )}  

où  ak , bk , ck   sont réels ou complexes et θ  variable réelle (angle exprimé en radians) 
Ces polynômes et séries sont  périodiques de période 2π

θ  est souvent noté w t  où w= 2π
T

  avec T>0  et t variable réelle (souvent le temps)

 ⇒ T est alors la période des séries et polynômes trigonométriques

Propriétés des polynômes trigonométriques
Tout polynôme trigonométrique de degré n et période T peut s’écrire :

Pn ,w( t)=∑
k=−n

n

c k ei k w t=∑
k=0

n

{ak cos(k w t )+bk sin (k w t)}   avec  w= 2π
T

 , T > 0  vérifiant :

     (tous les coefficients sont réels ou complexes)

                                                    c k=
1
T
∫

p

q

Pn ,w (t) e−i k w t dt                   avec  q-p=T                     

a0=
1
T
∫
p

q

Pn , w(t )dt                  a k=
2
T
∫

p

q

Pn ,w (t)cos (k w t)dt  

bk=
2
T
∫

p

q

Pn ,w (t)sin (k w t)dt

c k=
1
2
(ak−i bk)      c−k=

1
2
(ak+ i bk)        a0=c0        ak=(c k+c−k)        bk=i(ck−c−k)

                          1
T
∫
p

q

|Pn ,w (t)|
2 dt=∑

k=−n

n

|c k|
2=|a0|

2+ 1
2
∑
k=0

n

(|a k|
2+|bk|

2)                           

Si  Qn , w(t )=∑
k=−n

n

γ k ei k w t=∑
k=0

n

{α k cos (k w t)+β k sin(k w t)}  

 est un 2e polynôme de degré n et de même période T :

                         1
T
∫
p

q

Pn , w(t )Qn ,w (t)dt=∑
k=−n

n

ckγ k=a0α 0+
1
2
∑
k=0

n

(akα k+bk β k)



Propriétés des séries trigonométriques
On dit qu’une propriété est vraie pp (presque partout) si elle est vraie sauf en un nombre de 
points fini ou infini dénombrable.
Soit  f(t) une  fonction  réelle  ou  complexe  de  la  variable réelle t, périodique  de période     
T > 0, intégrable et de carré intégrable sur un intervalle quelconque [p,q] de longueur T, 
continue pp ainsi que sa dérivée première, sans asymptote verticale.
On associe à f(t) une série trigonométrique fSF(t) baptisée «Série de Fourier de f(t)» :

fSF (t)=∑
k=−∞

∞

ck ei k w t=∑
k=0

∞

{ak cos (k w t)+bk sin(k w t )}   avec  w= 2π
T

 , T > 0  vérifiant :

     (Tous les coefficients sont réels ou complexes)

                                              c k=
1
T
∫

p

q

f ( t)e−i k w t dt                   avec  q-p=T                                

a0=
1
T
∫
p

q

f ( t)dt                  a k=
2
T
∫

p

q

f (t )cos(k w t )dt               b k=
2
T
∫

p

q

f (t )sin(k w t )dt

c k=
1
2
(ak−i bk)     c−k=

1
2
(ak+ i bk)     a0=c0     b0=0     ak=(c k+c−k)     bk=i(ck−c−k)

              (Phénomène de Gibs démontré par Dirichlet)   
Pour tout t ϵ [p,q] on a fSF(t)=f(t)                         si f continue au temps t               
                                     fSF(t)=(1/2).{f(t+)+f(t-)}   si f discontinue au temps t

 

 1
T
∫
p

q

|f ( t)|2 dt= 1
T
∫

p

q

|fSF ( t)|2 dt=∑
k=−∞

∞

|ck|
2=|a0|

2+ 1
2
∑
k=1

∞

(|ak|
2+|bk|

2)    (Parseval)                       

Si  gSF(t )=∑
k=−∞

∞

γ k ei kw t=∑
k=0

∞

{α k cos (k w t)+β k sin(k w t)}  

 est la série de Fourier d’une 2e fonction g de même période T :  

 1
T
∫
p

q

f ( t)g(t )dt= 1
T
∫

p

q

fSF ( t)gSF(t)dt=∑
k=−∞

∞

ckγ k=a0α 0+
1
2
∑
k=1

∞

(akα k+bk β k)   (Parseval)

 On utilise ici les notations suivantes (avec ϵ>0) :
 f(t+)= lim

ε →0
f (t+ε )    et   f(t-)= lim

ε →0
f ( t−ε )  ; z =conjugué du nb complexe z

Remarque : Dans le cas d’une Série de Fourier on a n=∞ ce qui génère des problèmes de 
convergence. Les conditions d’intégrabilité et de continuité imposées à f(t) et f’(t) impliquent 
que fSF(t) représente bien la fonction f(t) (hors d’un nombre fini ou infini dénombrable de 
points de discontinuité situés dans un intervalle [p,q] de longueur T ) .                                        
Ce sont des conditions suffisantes. On ne connaît pas de conditions nécessaires et 
suffisantes.
                                                              



Noyaux de Dirichlet
Définitition :

Dn , w(t)=
1
T
∑

k=−n

n

e− i k w t    w=2π
T

    T > 0      n entier ≥ 0

                                                                                                                                                   
Propriétés :

Dn , w(t)=
1
T
(
sin ((n+1

2
)w t)

sin (1
2

w t)
)     ∫

−T /2

+T /2

Dn ,w (t)dt=1     Dn , w(0)=
2 n+1

T

Noyaux de Fejèr
Définition :

K n ,w( t)=
1

(n+1)∑k=0

n

Dk ,w (t)   (Moyenne de Césaro des n+1 premiers noyaux de Dirichlet)

Propriétés :                                                               

K n ,w( t)=
1

(n+1)T ( sin (
(n+1)

2
w t )

sin (1
2

w t ) )
2

    ∫
−T /2

+T /2

K n ,w (t)dt=1     K n ,w(0)=
n+1
T

    K n ,w( t)≥0  

Approximations :                                      

Soit fSF (t)=∑
k=−∞

∞

ck ei k w t=a0+∑
k=1

∞

{ak cos(k w t)+bk sin (k w t )}  la Série de Fourier de la 

fonction f(t). On nomme 
« Série de Fourier-Dirichlet limitée à l’ordre n de f(t) » le polynôme trigonométrique 

fSFDn( t)=∑
k=−n

n

ck ei k w t=∑
k=0

n

{ak cos (k w t)+bk sin (k w t )}

et « Série de Fourier-Fejèr limitée à l’odre n de f(t) » le polynôme trigonométrique 

fSFFn(t)=
1

n+1∑k=0

n

fSFDk( t)  

(c’est la moyenne de Césaro des n+1 premiers fSFDk(t))

Propriétés des approximations :

fSFDn( t)=∫
p

q

f (x)Dn(t−x)dx=∫
p

q

f ( t−x)Dn(x)dx         q-p=T   

fSFFn(t)=∫
p

q

f (x)K n( t−x)dx=∫
p

q

f (t−x)Kn(x)dx   

On reconnaît la convolution de fonctions de période T=q-p sur l’intervalle [p,q] : 
fSFDn=f ∘Dn     et    fSFFn=f ∘Kn                                              



II ) Th1 :
f et f’ continues pp dans [p,q]  
 de période T=q-p
intégrables
de carrés intégrables

  

 ⇒

 Parmi les polynômes trigonométriques de degré au plus N,
 fSFN(t) est le plus proche de f(t) au sens des moindres carrés
 (aN) et (bN) ont pour limites 0 quand N→∞

 ZN=∫
p

q

{fSF N (t )−f (t )}2 dt  décroissante et de limite 0 si N→∞ 

Démonstration : on utilise les formules élémentaires suivantes   ( où  w=2π/T )

(1) ∫
p

q

cos (n w t)dt=∫
p

q

sin (n w t )dt=0         , ∀n∈ℕ  ,  n > 0

(2) ∫
p

q

sin (n1 w t)cos (n2 w t)dt=0               , ∀n1 , n2∈ℕ   

(3) ∫
p

q

sin (n1 w t)sin (n2 w t)dt=∫
p

q

cos (n1 w t )cos (n2 w t)dt=0   , ∀n1 , n2∈ℕ  ,  0 < n1 ≠ n2 > 0

(4) ∫
p

q

sin2(n w t)dt=∫
p

q

cos2(n w t)dt=T
2

  , ∀n∈ℕ  ,  n > 0

On cherche le polynôme trigonométrique  PN (t)=a0+∑
n=1

N

{ancos (n w t)+bn sin(n w t)}  qui

possède 2N+1 coefficients et minimise  ZN=∫
p

q

{PN( t)−f (t)}2 dt  c’est-à-dire :

0=
∂ZN

∂ a0

=2∫
p

q

{PN ( t)−f (t )}dt=2{a0 T−∫
p

q

f (t)dt }    ⇒ a0=
1
T
∫
p

q

f (t)dt

0=
∂ZN

∂ an

=2∫
p

q

{PN (t)−f ( t)}cos (n w t )dt=2 {an
T
2
−∫

p

q

f ( t)cos (n w t)dt }  

 ⇒ an=
1
T
∫

p

q

f (t )cos(n w t)dt

0=
∂ZN

∂ bn

=2∫
p

q

{PN (t)−f ( t)}sin (n w t )dt=2{bn
T
2
−∫

p

q

f (t)sin (n w t )dt }

 ⇒ bn=
1
T
∫

p

q

f (t )sin(n w t)dt

  Il est donc démontré que PN(t) = fSFN(t)       cqfd   

                                                                                         

  La formule de Parseval 1
T
∫
p

q

|f ( t)|2 dt= 1
T
∫

p

q

|fSF ( t)|2 dt=∑
k=−∞

∞

|ck|
2=|a0|

2+ 1
2
∑
k=1

∞

(|ak|
2+|bk|

2)              

  prouve que la série de terme général (|ak|
2+|bk|

2) converge, donc ce terme général tend vers 0.
  On peut donc affirmer que, séparément, les suites (aN) et (bN) tendent vers 0.   cqfd
                                                                                                                           



  ZN−ZN−1=∫
p

q

{(fSF N (t)−f ( t))2−(fSF N−1(t)− f ( t))2}dt

=∫
p

q

{fSF N (t)− fSFN−1(t)}{fSF N (t)+ fSFN−1( t)−2 f ( t)}dt  

=∫
p

q

{aN cos(Nwt )+bN sin (Nwt )}{a0+{aN cos(Nwt )+bN sin (Nwt )−2 f (t)}+2 S}dt

                                                                             (    où S=2 ∑
n=1

N−1

{ancos (nwt )+bn sin(nwt )}   )

=∫
p

q

{aN
2 cos2(Nwt )+bN

2 sin2(Nwt )−2 aN f (t )cos (Nwt )−2 bN f (t)sin (Nwt )}dt

ZN−ZN−1=T {1
2

aN
2 + 1

2
bN

2 −aN
2 −bN

2 }=−T
2

{aN
2 +bN

2 }≤0   et la suite (ZN) est décroissante    cqfd

 ZN=∫
p

q

{fSF N (t )−f (t )}2 dt=∫
p

q

{fSF N (t)−fSF (t)}2 dt=∫
p

q

{ ∑
n=N+1

∞

{ancos (n w t)+bn cos(n w t)}}
2
dt

ZN=
T
2
∑

n=N+1

∞
{an

2+bn
2 }  c’est le reste pour la série convergente de Parseval. 

 Donc ZN tend vers 0 quand N tend vers l’∞   cqfd

 
  

 Th2 : Toute série trigonométrique S( t)=a0+∑
k=0

∞

{ak cos (k w t)+bk sin(k w t)}  où w=2π/T

   peut s’écrire   S( t)=r0+∑
k=1

∞

r k cos (k w t−φ k)=r 0+∑
k=1

∞

r k cos (k w (t−t k))     

                          S( t)=r0+∑
k=1

∞

r k sin (k w t−ψ k)=r 0+∑
k=1

∞

r k sin (k w (t−τ k))   avec

                               r0=a0   ,  φ0=ψ0=0  ,  r k=√ak
2+bk

2=
ak

cos (φ k)
=

bk

cos(ψ k)

                       tg(φ k)=
bk

ak

       tg(ψ k)=−
ak

bk

         φ k=k w t k        ψ k=k wτ k        

             φk et ψk sont définis à 2jπ près           j entier            tk et τk sont définis à 2jT près
                  Si  0≤φ k<2π   et    0≤ψ k<2π   ils sont nommés « déphasages-retard »
                  Si  0≤t k<T       et    0≤τ k<T      ils sont nommés « temps de retard »
                  De même pour tout polynôme trigonométrique (on remplace ∞ par N)

Démonstration : 

S( t)=a0+∑
k=1

∞

{ak cos(k w t)+bk sin (k w t )}=a0+∑
k=1

∞

ak {cos (k w t)+
bk

ak

sin (k w t )}

= a0+∑
k=1

∞

ak {cos (k w t )+tg(φ k)sin (k w t)}

= a0+∑
k=1

∞ ak

cos (φ k)
{cos (k w t)cos (φ k)+sin (φ k)sin (k w t )}=r0+∑

k=1

∞

r k cos (k w t−φ k)   cqfd



= r 0+∑
k=1

∞

r k cos(k w (t−t k))   cqfd

S( t)=a0+∑
k=1

∞

{ak cos(k w t)+bk sin (k w t )}=a0+∑
k=1

∞

bk {
ak

bk

cos (k w t)+sin (k w t)}

= r 0+∑
k=1

∞
bk {sin (k w t)−cos(k w t )tg (ψ k)}=r0+∑

k=1

∞ bk

cos (ψ k)
{sin(k w t )cos (ψ k)−sin(ψ k)cos (k w t)}

= r 0+∑
k=1

∞
r k sin (k w t−ψ k)=r 0+∑

k=1

∞
r k sin (k w (t−τ k))   cqfd



III ) SÉRIES DE FOURIER OBTENUES À PARTIR DE SÉRIES ENTIÈRES
                                                                           Période T                                                                                                  
w=2π/T ;  r et T réels>0 ; 0 ≤ s ≤ 1 ; p entier ≥ 0 

  ∑
n=0

∞
sncos (n w t)=

1−s cos(w t)
1+s2−2 s cos (w t)

        0 ≤ t ≤ T

  ∑
n=1

∞

snsin (n w t )=
ssin (w t)

1+s2−2 scos (w t )
        0 ≤ t ≤ T

  ∑
n=0

∞

(−1)n sncos (n w t)=
1+s cos(w t)

1+s2+2 s cos (w t)
                                                               0 ≤ t ≤ T

  ∑
n=1

∞

(−1)n+1 snsin (n w t )=
s sin (w t)

1+s2+2 scos (w t )
        0 ≤ t ≤ T

  ∑
n=0

∞ rn

n!
cos (n w t)=er cos(w t )cos (r sin (w t ))                                                                 0 ≤ t ≤ T

  ∑
n=1

∞ rn

n !
sin (n w t)=er cos (w t )sin (r sin (w t))         0 ≤ t ≤ T

  ∑
n=0

∞

(−1)n rn

n!
cos(n w t)=e−r cos(w t )cos (r sin (w t ))         0 ≤ t ≤ T

  ∑
n=1

∞

(−1)n+1 rn

n !
sin (n w t)=e−r cos(w t) sin(r sin(w t))         0 ≤ t ≤ T 

  ∑
n=0

∞ r2n

(2 n)!
cos (2 n w t)=ch(r cos (w t))cos(r sin (w t ))          0 ≤ t ≤ T

  ∑
n=1

∞ r 2 n

(2 n)!
sin (2 n w t)=sh(r cos (w t ))sin (r sin (w t))           0 ≤ t ≤ T

  ∑
n=0

∞

(−1)n r 2n

(2 n)!
cos (2 n w t )=ch(r sin(w t))cos (r cos (w t ))         0 ≤ t ≤ T

  ∑
n=1

∞

(−1)n+1 r 2 n

(2 n)!
sin (2 n w t)=sh(r sin (w t))sin (r cos(w t))         0 ≤ t ≤ T

  ∑
n=0

∞ r2 n+1

(2n+1)!
cos ((2n+1)w t)=sh(r cos (w t ))cos(r sin(w t))         0 ≤ t ≤ T

  ∑
n=0

∞ r2 n+1

(2n+1)!
sin ((2 n+1)w t)=ch(r cos (w t ))sin (r sin(w t))         0 ≤ t ≤ T

  ∑
n=0

∞

(−1)n r2n+1

(2 n+1)!
cos ((2 n+1)w t )=ch(r sin(w t))sin (r cos (w t ))

        0 ≤ t ≤ T

  ∑
n=1

∞

(−1)n r 2n+1

(2 n+1)!
sin ((2 n+1)w t )=sh(r sin (w t ))cos(r cos (w t))                                            

        0 ≤ t ≤ T 



                                                                           Période T
w=2π/T ; T réel>0 ; p entier ≥ 0 ; Bp(x) et Ep(x) polynômes de Bernoulli et Euler de degré p
                             

  ∑
n=1

∞ cos (n w t)
n2 p =(−1)p+1(22 p−1π 2 p

(2 p)!
)B2 p(

t
T
)                                                          0 ≤ t ≤ T   p entier > 0   

  ∑
n=1

∞ sin (n w t )
n2 p+1 =(−1)p+1(22 pπ 2 p+1

(2 p+1)!
)B2 p+1(

t
T
)

0 ≤ t ≤ T  si p > 0  
        0 < t < T  si p = 0 

  ∑
n=0

∞ cos ((n+1 /2)w t)
(2 n+1)2 p =(−1)p( π 2 p

4 (2 p−1)!
)E2 p−1(

t
T
)         0 ≤ t ≤ T et p > 0     

  ∑
n=0

∞ sin ((n+1 /2)w t)
(2 n+1)2 p+1 =(−1)p( π 2 p+1

4(2 p)!
)E2 p(

t
T
)

        0 ≤ t ≤ T  si p > 0  
        0 < t < T  si p = 0 

Démonstrations :

∑
n=0

∞
zn= 1

1−z
 si |z| < 1 ; si z=s ei w t

  ⇒ ∑
n=0

∞
snei w t= 1

1−s ei w t=
(1−scos (w t ))+ i ssin (w t)

1+s2−2 s cos (w t)
  ⇒

∑
n=0

∞

sncos (n w t )=
1−s cos(w t)

1+s2−2 s cos (w t )
 et ∑

n=1

∞

snsin (n w t )=
ssin (w t)

1+s2−2 scos (w t )
 cqfd   si s → -s  ⇒

∑
n=0

∞

(−1)n sncos (n w t)=
1+s cos(w t)

1+s2+2 s cos (w t)
 et ∑

n=1

∞

(−1)n+1 snsin (n w t )=
s sin (w t)

1+s2+2 scos (w t )
 cqfd

∑
n=0

∞ zn

n!
=e z   z complexe quelconque. Si z=reiwt   ⇒ ∑

n=0

∞ rn

n!
ei n w t=er(cos(w t)+i sin (w t))

  ⇒

∑
n=0

∞ rn

n!
(cos (n w t )+ i sin(n w t))=er cos(w t)(cos (r sin(w t))+ i sin (r sin(w t)))   ⇒

∑
n=0

∞ rn

n!
cos (n w t)=er cos (w t )cos (r sin (w t ))  et ∑

n=1

∞ rn

n !
sin (n w t)=er cos (w t )sin (r sin (w t))   cqfd             

si r → -r   ⇒ ∑
n=0

∞

(−1)n rn

n!
cos(n w t)=e−r cos(w t )cos (r sin (w t ))  et

∑
n=1

∞

(−1)n+1 rn

n !
sin (n w t)=e−r cos(w t)sin(r sin(w t))  cqfd

∑
n=0

∞ z2 n

(2 n)!
=ch(z)  z complexe quelconque Si z=rei w t  ⇒

∑
n=0

∞ r2 n

(2n)!
ei2n w t=ch(r ei w t)=ch(r cos (w t )+ i sin(w t))

∑
n=0

∞ r2n

(2 n)!
(cos (w t )+i sin(w t))=ch(r cos (w t))ch(i r sin (w t))+sh(r cos(w t)) sh(i r sin (w t ))



= ch(r cos (w t))cos(r sin (w t ))+ i sh(r cos (w t ))sin (r sin(w t))  ⇒

∑
n=0

∞ r2n

(2n)!
cos (2 n w t)=ch(r cos(w t))cos(r sin (w t))  et

∑
n=1

∞ r 2 n

(2 n)!
sin (2 n w t)=sh(r cos (w t ))sin (r sin (w t))  cqfd  si r → i r  ⇒

∑
n=0

∞

(−1)n r 2 n

(2 n)!
cos (2 n w t )=ch(r sin(w t))cos (r cos (w t ))  et

∑
n=1

∞

(−1)n+1 r 2n

(2 n)!
sin (2 n w t)=sh(r sin (w t))sin (r cos(w t))  cqfd

∑
n=0

∞ z2 n+1

(2n+1)!
=sh(z)   z complexe quelconque Si z=rei w t ⇒

∑
n=0

∞ r2 n+1

(2n+1)!
ei n w t=sh(r ei w t)=sh(r cos (w t )+i r sin (w t))

∑
n=0

∞ r2 n+1

(2n+1)!
ei n w t=sh(r ei w t)=sh(r cos (w t ))ch( i r sin(w t))+ch(r cos(w t)) sh(i r sin (w t))

= sh(r cos(w t))cos (r sin(w t))+i ch(r cos (w t ))sin (r sin(w t))  ⇒

∑
n=0

∞ r2 n+1

(2n+1)!
cos ((2n+1)w t)=sh(r cos (w t ))cos(r sin(w t))  et

∑
n=0

∞ r2 n+1

(2n+1)!
sin ((2 n+1)w t)=ch(r cos (w t ))sin (r sin(w t))  cqfd       si r → i r  ⇒

∑
n=0

∞

(−1)n r2n+1

(2 n+1)!
cos ((2 n+1)w t )=ch(r sin(w t))sin (r cos (w t ))  et

∑
n=1

∞

(−1)n r 2 n+1

(2 n+1)!
sin ((2 n+1)w t )=sh(r sin (w t ))cos(r cos (w t))  cqfd

Dans l’article « Nombres et Polynômes de Bernoulli et Euler » les 4 formules suivantes sont démontrées :
(les notations inversant les rôles de n et p)

 ∑
n=1

∞

(
cos(2 nπ x)

n2 p )=(−1)p+1(22 p−1π 2 p

(2 p)!
)B2 p(x)    si  0 ≤ x ≤ 1    p >0

 ∑
n=1

∞

(
sin(2nπ x)

n2 p+1 )=(−1)p+1( 22 pπ 2 p+1

(2 p+1)!
)B2 p+1(x)   si 0 ≤ x ≤ 1 si p > 0 ,  0 < x < 1 si p = 0



 ∑
n=0

∞

(
cos((2 n+1)π x)

(2 n+1)2 p )=(−1)p( π 2 p

4(2 p−1)!
)E2 p−1(x)   si p ≥ 1 et  0 ≤ x ≤ 1

 ∑
n=0

∞

(
sin ((2 n+1)π x)
(2 n+1)2 p+1 )=(−1)p( π 2 p+1

4(2 p)!
)E2 p(x)    si  p≥1 et 0≤ x ≤1

p=0 et 0<x<1
 

 Posons x=t/T   ⇒ πx=wt/2 il vient : ∑
n=1

∞ cos (n w t)
n2 p =(−1)p+1(22 p−1π 2 p

(2 p)!
)B2 p(

t
T
)

∑
n=1

∞ sin (n w t )
n2 p+1 =(−1)p+1(22 pπ 2 p+1

(2 p+1)!
)B2 p+1(

t
T
)

∑
n=0

∞ cos ((n+1 /2)w t)
(2 n+1)2 p =(−1)p( π 2 p

4 (2 p−1)!
)E2 p−1(

t
T
)

∑
n=0

∞ sin ((n+1 /2)w t)
(2 n+1)2 p+1 =(−1)p( π 2 p+1

4(2 p)!
)E2 p(

t
T
)  cqfd

                                                 

   



IV ) PREMIERS EXEMPLES
  

           f1=:1r4p1**
           [AB1=. (f1 SF  _5  5) 9
0 0 0 0              0 0    0 0              0 0              0 0                0
0 1 0 0.333333 0 0.2 0 0.142857 0 0.111111 0 0.0909091 0
           f1s=:(AB1 FPT _5 5)"0
           'pensize 3' plot (];f1,:f1s)1000 sint _5 5

 Mise en évidence du phénomène de Gibbs

       f2=: _5  5&{{1r8p2-1r4p1*w*y**y[w=.2p1%T=.-/|.x}}"0 0

       [AB2=. (f2 SF  _5  5)8
0 1 0 0.111111 0 0.04 0 0.0204082 0
0 0 0 0             0 0      0 0                0
       f2s=:(AB2 FPT  _5  5)"0
       'pensize 3' plot (];f2,:f2s)1000 sint _5 5

  f3=:14&{{(1r8p2*w*y)-((1r8p1**:w*y)**y)[w=.2p1%T=.x}}"0 0

  [AB3=. (f3 SF   _5   9)6
0 0 0 0              0 0        0 0                  0
0 1 0 0.037037 0 0.008 0 0.00291545 0
  f3s=:(AB3 FPT   _5  9)"0
  'pensize 3' plot (];f3,:f3s)1000 sint _5 9

 phénomène de Gibbs aux extrémités de l’intervalle

     f4=:10&{{(1r96p4-*:1r4p1*w*y)+(*y)*1r24p1*(w*y)^3[w=.         
2p1%T=.x}}"0 0
     [AB4=. (f4 SF _5  5) 8
0 1 0 0.0123457 0 0.0016 0 0.000416493 0
0 0 0 0                0 0          0 0                    0
     f4s=:(AB4 FPT  _5  5)"0
     'pensize 3' plot (];f4,:f4s)1000 sint _9 2

  intervalle de définition [-5  5]
  intervalle d’affichage   [-9  2]

                                                                                                                                                                                      

                                                                        



V ) EXEMPLES PRATIQUES 



 

Puissance d’une fonction f sur un intervalle [p,q)
        f =: ^*1&o.
        f   PUISSANCE   1  5
1552.44              



       ff =: ^                                NB. ex                  
       gg =: 1 1 1r2&p.              NB. 1+x+0,5x2     série de Mac Laurin ordre 2
       (<'ff')   COMPARAISON   (<'gg')    _2  3
11.5855   1.51084   2.99699

 

      ff=:^                                                            NB. ex 
      AB=. (ff SF _2 3) 2                                     NB. Coeff. série de Fourier ordre 2
      ffSF2 =: (AB FPT _2 3)"0                         NB. Fonction série de Fourier
      (<'ff') COMPARAISON (<'ffSF2') _2 3
11.9107   1.80744   2.77906
                                                    
   Retourne les pseudo-distances  d’ordre 0 (Max |f-fSF2|) sur [_2   3]
  d’ordre 1 (intégrale de |f-fSF2| sur [_2   3])%5          5=longueur de l’intervalle
  d’ordre 2 (intégrale de |f-fSF2|2 sur [_2   3])1/2%5                                     

       h=:2:+(3:*]>1:)-(7:*]>3:)
       M=: (h SF 0 5) 3
       hSF3=: (M FPT 0 5)"0
       (<'h') COMPARAISON (<'hSF3') 0  5
3.53664   0.85294   1.11995
 

                                                                                                                                                       

      ABe                                                        
0.6 0.32 0.88 0.04 0.36 0.32 0.08 0.6   0.28 0.92
0    0.4   0.68 0.8   0.76 0.44 0.6   0.16 0.72 0.16
      [ABs=.5 FPB ABe             NB. filtre passe-bas
0.6 0.32 0.88 0.04 0.36 0.32
0    0.4   0.68 0.8   0.76 0.44
      fe=:(ABe FPT _1 6)"0
      fs=:(ABs FPT _1 6)"0
      (<'fe') COMPARAISON (<'fs') _1 6 
2.65161 0.804069 1.05376
 

 L’application d’un filtre passe-
bas est une méthode de lissage

      ABe
0 0.1 2      0.1 3 8 4 0.2 _0.3
0 0.2 0.3 _0.1 5 7 2 0.1   0.5
      [ABs=. 4  6  FPBD  ABe         NB. Filtre passe-bande
0 0 0 0 3 8 4 0 0
0 0 0 0 5 7 2 0 0
      fs=: ABs FPT 2 7"0   [.   fe=: ABe FPT 2 7"0
      (<'fe')  COMPARAISON  (<'fs')  2  7
2.76437 1.3135 1.50831



      f =: 5*^@((1&o.)+(2&o.)@(9&*))
      g =: (0 9;10)&( (4&FPB)  FILTRE  f )"0     NB. filtre passe-bas
      (<'f')  COMPARAISON  (<'g')  0 9
19.0721 5.45928 7.06478

 La conjonction FILTRE applique directement à f (en passant implicitement par 
les matrices de coefficients) le filtre passe-bas (4&FPB) pour obtenir la fonction 
lissée g. Sur un intervalle de longueur égale à la période la fonction lissée g a 
même intégrale que f.           Lissage

      f =: 5*^@((1&o.)+(2&o.)@(9&*))
      h =: (0 9;10)&( (5&FPH)  FILTRE  f )"0     NB. idem filtre passe-haut
      (<'f')  COMPARAISON  (<'h')  0 9
35.3732 9.49499 12.8728

 

      ABe
0 0 0 0 0 0 0 0 2   3 4 1
0 0 0 0 0 0 0 0 1 _1 2 3
      [ABs =. _7  FFREQ   ABe     NB. Baisse de 7 harmoniques
0 2   3 4 1
0 1 _1 2 3
      fs=: ABs FPT 1 5"0    [.   fe=: ABe FPT 1 5"0
      (<'fe')   COMPARAISON   (<'fs')  1  5
22.4789 4.61855 6.7082

      X=.1 2 5 9 10 16  [  Y=.4 8 2 3 6 7

      [ AB=.  (X;Y)   XYSFPOL   (0 20;7)    NB. Fonction polynômiale
5.2398 _1.5486 _1.1601 _0.5792 _0.3383 _0.2202 _0.1544 _0.1140
0          _3.2788   4.3718   0.6697 _0.2523 _0.5045 _0.5626 _0.5572
      f =: (AB   FPT   0 20)"0 
      pd  'title Série de FOURIER à partir de points'
      pd   'type line ; pensize 2 ; color blue'
      pd   (];f)1000 sint 0 20
      pd   'type marker ; color red ; markersize 2'                                                    
      pd   X;Y
      pd   'show'                                                                                                             
 La courbe n’est connue que par un nombre fini (n+1) de points (X;Y) . Elle est 
remplacée par la courbe polynômiale de degré n passant par ces (n+1) points.
C’est cette fonction polynômiale qui est développée en série de Fourier.

                                                                                                                                                                           
 



    X=.0 1 5 9 12 13 15   [   Y=.2 5 6 4 1 0 _2

     f =: (X;Y) FESC "0          NB. Fonction en escalier
     AB =.  (f SF 0 15) 8           NB. Coeff. série de Fourier ordre 8
     g =: (AB FPT 0 15)"0       NB. Fonction série de Fourier
     pd 'new ; title Série de FOURIER à partir de points<br/>en passant par 
une fonction en escalier'
     pd 'type line ; pensize 3 ; color blue'
     pd (];f) 1000 sint 0 15
     pd 'type line ; pensize 3 ; color green'
     pd (];g)1000 sint 0 15
     pd 'type marker ; color red ; markersize 2' 
     pd X;Y
     pd 'show'

  

     f0=._1000 2 _3 0.4&p.

     f1=.1200 4 0 _0.5&p.
     f2=.880 * 1&o.
     f3=.100*]*(2&o.)@(2&*)
     X=.0 1 5 6 11 15 18 20
     Y=.700 500 400 400 600 900 800 700
                  (f0`f1`f2`f3)  AFFXY  (X;Y)  0 20

 forme monadyque

'COURBES ET POINTS'  (f0`f1`f2`f3)   AFFXY  (X;Y)  0 20

forme dyadique
avec un titre et indication des fonctions 

  X=.0 1 3 4 7 9 10 [ Y=.4 7 3 6 7 4 4
  fp =: (X POLXY Y)&p.     NB. Fonction polynômiale
       AB=. (X;Y) XYSFPOL (0 10;4)      NB. et
  fpSF=:(AB FPT 0 10)"0    NB. sa série de Fourier (ordre 4) 
  fe=: (X;Y) FESC"0            NB. Fonction en escalier
       ab=. (X;Y) XYSFESC (0 10;4)        NB. et
  feSF=:(ab FPT 0 10) "0     NB. sa série de Fourier (ordre 4) 
  titre=.'Séries de Fourier issues de points '
  titre=.titre,'isolés <br/> via des fonctions'
  titre=.titre,' polynômiale et en escalier'
  titre   (fp`fpSF`fe`feSF)   AFFXY  (X;Y)   0 10                        



VI) SERIES DE FOURIER : script
NB. SERIESDEFOURIER.ijs                                                                                                                
require 'plot'
require '~user/BOITAOUTILS.ijs'                                                                                                                                       
require '~user/BERNEULER.ijs'

NB. Coefficients de la série de Fourier d'une fonction f sur un intervalle p,q 
NB. de période T=q-p, limitée à un polynome trigonométrique de degré N (entier >: 0)
NB. AB : coeff. (matrice de 2 lignes et N+1 colonnes)
NB. 1ere ligne : coefficients des cosinus
NB. 2eme ligne : coefficients des sinus

NB. Utilisation :     AB =. ( f   SF  p,q ) N
SF =: {{)c  

wbxw=.1p1*nu2=.2*nu=.%T=.--/n [I=.1+i.N=.y
a=.nu*100 u. INTGLC n [b=.0
for_i. I do.
ftucos=.u.*((2&o.)@((i*wbxw)&*))
ftusin=.u.*((1&o.)@((i*wbxw)&*)) 
a=.a,nu2*100 ( ftucos INTGLC ) n
b=.b,nu2*100 ( ftusin INTGLC ) n
end. 
1e_13 AR0 a,:b }}

NB. Coefficients de la série de FOURIER (approximation)
NB. définie sur l'intervalle [p q] limitée à l'ordre N
NB. d'une fonction passant par les points (X ; Y)
NB. via une fonction polynômiale

NB. Utilisation : AB=. (X;Y) XYSFPOL (p q;N) 
XYSFPOL=:{{)d

'X Y'=.x [ 'pq N'=.y
C=.X POLXY Y
ff=.C&p.
(ff SF pq) N }}

NB. Coefficients de la série de FOURIER (approximation)
NB. définie sur l'intervalle [p q] limitée à l'ordre N
NB. via une fonction en escalier
NB. d'une fonction passant par les points (X ; Y)

NB. Utilisation : AB=. (X;Y) XYSFESC (p,q;N) 
XYSFESC =:{{)d

'pq N'=.y
ff=. x FESC "0
(ff SF pq) N }}
 

NB. Ft polynôme trigo de degré N, de coefficients AB et période T=q-p
NB. AB : matrice de 2 lignes (coef. de sin et cos) et N+1 colonnes
NB. Définition sur l'intervalle p,q

NB. Utilisation :    f =: (AB  FPT  p,q)"0 
FPT =:{{)c

I=.i.{:$AB=.m [ w=.2p1%T=.-/|.n
+/+/AB*2 1 o./I*w*y }}



NB. Ft polynôme trigo de degré N,de type cosinus, de coef. Ro, 
NB. de période T=q-p, de déphasages FIC (Ro et FIC vect de N+1 composantes
NB. Définition sur l'intervalle p,q

NB. Utilisation :     fptc =: ((Ro  ,: FIC)  FPTC  p,q)"0 
FPTC =:{{)c

Ro=.0{m [ FIC=.1{m [ w=.2p1%T=.-/|.n
+/Ro*2 o.FIC+(i.#Ro)*w*y }}

NB. Ft polynôme trigo de degré N,de type sinus, de coef. Ro, 
NB. de période T, de déphasages FIS (Ro et FIS vect de N composantes

NB. Utilisation :     ftps =: ((Ro ,: FIS)  FTPS  T)"0 
 FTPS =:{{)c
Ro=.0{m [ FIS=.1{m [ w=.2p1%T=.-/|.n
+/Ro*1&o.FIS+(i.#Ro)*w*y }}                                                                                                     

NB. Affichage pour t entre p et q de 2 (ou plusieurs) fonctions
NB. x ( facultatif ) : chaîne de car avec options de plot
NB. Utilisation : 

NB.    (f ` g   AFF)  p,q    ou    x  (f ` g   AFF)  p,q
NB.    (f0`f1`f2`f3`f4`f5   AFF)  p,q     ou      x (f0`f1`f2`f3`f4`f5   AFF)  p,q

AFF =:{{)a
z=. plot (];m`:0) 100 sint y
:                         
z=. x plot (];m`:0) 100 sint y }}

NB. Affichage pour t entre p et q de 1 à 4 fonctions
NB. et de points isolés (X;Y)
NB. Utilisation :     (f0`f1`f2`f3)   AFFXY   (X;Y)  (p,q)
NB. ou                   'TITRE'  (f0`f1`f2`f3)   AFFXY   (X;Y)  (p,q)

AFFXY=:{{)c
'type line ; pensize 3' plot (];(m`:0)) 1000 sint y
pd 'type marker ; color black ; markersize 2'
pd n
pd 'show'
: 
op1=.'type line;pensize 3;titlefont Arial 17 bold italic;title ',x
op2=.';color blue,red,green,purple;keypos lto;key ',(}.,',',.>m)
op3=.';keycolor blue,red,green,purple' 
(op1,op2,op3) plot (];(m`:0)) 1000 sint y 
pd 'type marker ; color black ;markersize 1'
pd n
pd 'show' }} 

 



NB. Analyse de Fourier d’une fonction f limitée à l’ordre N
NB. [p,q] : intervalle de définition ;  période T=q-p

NB. Utilisation :        R =.  ( (<'f')   ANALYSEDEFOURIER   p,q )  N
ANALYSEDEFOURIER =: {{)c

z=. P 3 
z=. (((>m);((>m),'SF',(":y))) GRAPHIQUES n) ZZ=. (m ANAF n) y
'AB Ro FIC FIS Pf Pah Prh Pad Prd'=:ZZ 
M=. AB,Ro,FIC,FIS,Pah,Prh,Pad,:Prd
M1=.(<' hd'),:<' a',' b',' Ro','FIC','FIS','Pah','Prh','Pad',:'Prd'
M2=. M1,.((1 1,8$0);(1)) MatBloc (i.y+1),5e_4 AR0 M
(Pf(m Titre n)y),(<M2),:Notice }}                                                                                                                      

NB. Pseudo-distance quadratique entre 2 fonctions f et g 
NB. Sur un intervalle fini [p,q]

NB. Utilisation :    R =. (f  psdist2  g)   p,q
psdist2 =: {{ %:(-/|.y)%~100 (((*:@(u-v))"0) INTGLC) y}}

 NB. Pseudo-distance en valeur absolue entre 2 fonctions f et g

NB. Sur un intervalle fini [p,q]

NB. Utilisation :    R =. (f  psdist1  g)  p,q
psdist1 =: {{(-/|.y)%~100 (((|@(u-v))"0) INTGLC) y}} 

NB. Pseudo-distance en valeur maxi entre 2 fonctions f et g
NB. Sur un intervalle fini [p,q]

NB. Utilisation :    R =. (f  psdist0  g)  p,q
psdist0 =: {{>./((|@(u-v))"0)(1000&sint) y}}

NB. Comparaison de 2 fonctions f1 et f2 sur un intervalle [p,q]
NB. avec affichage et des courbes et pseudo-distances: R = Dist0, Dist1, Dist2
NB. Dist0 : distance de la convergence uniforme
NB. Dist1 : moyenne de |f1-f2| sur [p,q] 
NB. Dist2 : racine carrée de la moyenne du carré de (f1-f2) sur [p,q]

NB. Utilisation :   R =. (<'f1')  COMPARAISON  (<'f2')  p,q
COMPARAISON =: {{)c

'D0 D1 D2'=.((u1 psdist0 v1),(u1 psdist1 v1),((u1=.m`:6) psdist2 (v1=.n`:6)))y
d0=.0j3 ": D0 [ d1=.0j3 ": D1 [ d2=.0j3 ": D2 
CAP=.';xcaption Dist0=',d0,' Dist1=',d1,' Dist2=',d2 
op=.options,(>m),' ',(>n),,CAP
op plot (];u1,:v1) 100 sint y

D0,D1,D2 }}

NB. Fonctions série de Fourier d'une fonction f pour t compris entre p et q
NB. correspondant aux multiples degrés d0 d1 d2...dk
NB. création des fonctions fSFd0, fSFd1,...fSFdk
NB. dont la liste emboîtée est dans R 

NB. Utilisation :   R =. ((<'f')   SFM   p,q)    d0 d1 d2...dk
SFM=:{{)c

N=.>./y [T=.-/|.n [R=.m
CC =. ( (m`:6) SF n ) N
for_i. y do. ii=._2{.'0',":i
dd=.((i+1)&{.)&.|:CC



R=.R,<X=.'SF',ii,(>m)
". X,'=:(dd FPT n)"0'
end.
R }}

NB. Puissance d’une fonction f sur un intervalle [p,q] 
NB. Utilisation : R =. (f   PUISSANCE)  p,q

PUISSANCE =: {{)a
UUU=:(*:@u)"0
R=.(1000(UUU INTGLC) y)%-/|.y 
poub=.EFFACE <'UUU'
R }}
 

NB. Produit de convolution de 2 fonctions (conjonction)
NB. f et g sur un intervalle [p,q]            h=f°g

NB. Utilisation :        h =: ((f `g)   CONVOLUTION   p,q)"0 
CONVOLUTION =: {{100 ((({.m)`:6)*(((({:m)`:6)NMY y)"0))INTGLC n}}

NB. Méthode des noyaux
NB. Noyaux de Dirichlet NDIR
NB. T : période N : degré
NB. (NDIR 0)=(1+2N)/T 
NB. Intégrale de NDIR dans [-T/2,T/2] = 1
NB. Développement en série de Fourier 
NB. à l'ordre N de f(t) de période T est
NB. SFf(t) =: intégrale de _T/2 à T/2 de
NB. f(x).NDIR(t-x)*dx 

NB. Utilisation :      NDIR  =:  (N  DIR  T)"0   
DIR =: {{)c

N=.m [ T=.n 
if. ( 1e_13 >|y) do. R=.(1+2*N)%T
else. R=. (1&o.(1+2*N)*1p1*y%T)%T*1&o.1p1*y%T end.
R }}

NB. Noyaux de Fejér NFEJ 
NB. (sommes de Césaro des noyaux de Dirichlet))
NB. T : période N : degré
NB. (NFEJ 0)=(1+N)/T
NB. Intégrale de NFEJ dans [-T/2,T/2] = 1
NB. Développement en série de Césaro-Fourier 
NB. à l'ordre N de f(t) de période T est
NB. SCFf(t) =: intégrale de _T/2 à T/2 de
NB. f(x).NFEJ(t-x)*dx

NB. Utilisation :          NFEJ =: (N  FEJ  T)"0
FEJ =: {{)c

N=.m [T=.n
if. ( 1e_13 >|y) do. R=.(1+N)%T
else. R=.(*:(1&o.(1+N)*z)%1&o.z=.y*1p1%T)%T*1+N end.
R }}
           

            



NB. ************************************

NB. FILTRES
NB. Modifient la matrice des coefficients
NB. sans changer la période ni l’intervalle de définition 

NB.     ABe : entrée ;        ABs : sortie

NB. Utilisation :     ABs =. x  F….    ABe  
NB. ************************************

NB. FILTRES agissant sur les amplitudes

NB. Passe-bas                                                                                                        

NB. Utilisation : ABs =. k0  FPB  ABe
FPB =: {{|:(>:x){.|:y}}

NB. Passe-haut
NB. Utilisation :  ABs=. k0  FPH  ABe

FPH =: {{((2,x)$0),.|:x}.|:y}}

NB. Passe-bande
NB. Utilisation : ABs=. (k0,k1) FPBD ABe

FPBD=:{{0(<0 1;(i.0{x),(>:1{x)+i.(<:1{$y)-1{x)}y}}

NB. Filtre écrèteur
NB. Utilisation : ABs=. max  FECR  ABe 

FECR =: {{y*"1 r%~x<.r=.%:+/*:y}}

NB. Filtre plancher
NB. Utilisation : ABs=. min  FPL  ABe

FPL =: {{y*"1 r%~x>.r=.%:+/*:y}}

NB. Filtre général sur les amplitudes
NB. Utilisation : ABs=. x FG ABe
NB. ABs=. (((#|:ABe )$0 1) FG ABe NB. Sélection des harmoniques impaires 

NB. ABs=. (((#|:ABe )$1 0) FG ABe NB. Sélection des harmoniques paires 

NB. ABs=. k FG ABe NB. k réel >0 ; amplification ou atténuation 

NB. ABs=. k FG ABe NB. k réel <0 ; idem en opposition 

NB. ABs=. V FG ABe NB. (#V)=(#|:ABe) ; idem sélectivement  
FG =: {{x*"1 y}}

 NB. Filtre dérivateur
NB. Utilisation : ABs =. T FDER ABe 

FDER=: {{(2p1%m)*|. _1 1* (i.1{$y)*"1 y}}"0 _ 

NB. Filtre intégrateur (avec sortie nulle pour t=0)

NB. Utilisation : ABs =. T FINT ABe 
FINT =: {{)d

N=.<:1{$y [ a0=.0{0{y [ w=.2p1%x
R=. (y*(0,N$1),:(0,N$_1))
R=.R+(0,2*a0*N$1 _1),:0
R=.|.R(%"1)1,w*1+i.N
(-+/0{R)(<0;0)}R }}



NB. Filtre agissant sur les fréquences
NB. avec annulation de la composante constante a0

NB. x : entier ; ABe : matrice d’entrée ; ABs : matrice de sortie
NB. x=0 annule a0 ; x>0 décalage vers les hautes fréquences
NB. x<0 décalage vers les basses fréquences avec perte de coefficients

NB. Utilisation : ABs =. x  FFREQ    ABe
FFREQ =: {{)d

select. *x
case. 0 do. R=.0(<0 1;0)}y 
case. _1 do. R=.0(<0 1;0)}|:(|x)}.|:y 
case. 1 do. R=.((2,x)$0),.0(<0 1;0)}y end.
R }}

NB. Fonction fs obtenue par application d’un filtre x&F…. sur une fonction fe 
NB. Pour fe : ordre maxi N ; intervalle de définition [p q] 

NB. Utilisation : fs =: (p q;N) & ( ( x&F….)  FILTRE   fe )"0
 FILTRE =:{{)c
'pq N'=.x
ABe=.(v SF pq) N
ABs=.u ABe
(ABs FPT pq) y }}

       



NB. ************************************

NB. Outils divers

NB. ************************************
E =: 1!:2&2                                         NB. Affiche et transmet son argument de droite (monadique)
P =: 9!:11                                             NB. Fixe le nombre de chiffres affichés (monadique)                  

NB. Efface des objets 

NB. Utilisation :   EFFACE <'obj1'   ou    EFFACE 'obj1';'obj2';'obj3';...
EFFACE =: 4!:55

NB. Découpage de l’intervalle p,q en N sous-intervalles d'égales longueurs
NB. Utilisation :    R =. N  sint  p,q

sint =: {.@]+-/@|.@]*i.@>:@[%[
                                                                                                                                            

NB. Arrondi à 0 les éléments de y de valeur absolue < x (x>0) 
NB. Utilisation :         R =. x  AR0  y

AR0 =: (([<|@])*])"0 0

NB. Intégration numérique méthode de Gauss-Legendre composite 
NB. utilisant le polynôme de LEGENDRE de degré 7 
NB. f  (fonction à intégrer); p,q (intervalle d'intégration) divisé en n sous-intervalles de même longueur 

NB. Utilisation :          R =. n   f  INTGLC p,q
INTGLC =: {{)a

msi=.(>@(2&(<\))) x ({.@]+i.@>:@[*({:-{.)@]%[) y
LEGr7=._0.949107912342759 _0.741531185599395 _0.405845151377397
LEGr7=.LEGr7, 0,-|.LEGr7
LEGp7=.0.0647424830844263 0.139852695744671 0.190915025252494 
LEGp7=.LEGp7, 0.208979591836819, |.LEGp7 
VR=.i.0 [ I=.i.x
for_i. I do. VR=.VR,+/LEGp7*u. LEGr7 {{0.5*(+/y)-x*(-/y)}}"0 _ i{msi end.
(+/VR)*(--/y)%x }}

NB. Intégration numérique méthode de Romberg améliorant la méthode des trapèzes
NB. f (fonction à intégrer); p,q (intervalle d'intégration) 
NB. divisé en n sous-intervalles de même longueur (n : multiple de 8)

NB. Utilisation :       R =. n   f   INTROMB   p,q
INTROMB =:{{)a

h=.8*(b-a)%x [ 'a b'=.y [N=.x+1
X=.--:({.+{:)Y=.u."0 x sint y
Z=.(h*X++/(N$1,7$0)#Y),(-:h*X++/(N$1,3$0)#Y)
Z=.Z,((h%4)*X++/(N$1 0)#Y),((h%8)*X++/Y)
Z+/ .*_1r2835 4r135 _64r135 4096r2835 }}

NB. Série de Fourier d'ordre N d'une fonction f 
NB. sur un intervalle p,q , de période T=q-p
NB. retourne une matrice R de 7 lignes et N+1 colonnes
NB. ligne 0 (a) ; ligne 1 (b) ; ligne 2 (ro) 
NB. ligne 3 (déphasage retard pour cos) ; ligne 4 (temps retard pour cos)
NB. ligne 5 (déphasage retard pour sin) ; ligne 6 (temps retard pour sin)

NB. Utilisation :     R =. (f  SFRD  p,q) N



SFRD =: {{)c
X=.(u SF n )y [T=.-/|.n NB. coeff. série de Fourier en cos et sin
Y=.|:*.j./X NB. modules avec déphasage en cos dans ]_1p1 , 1p1]
dc=. _2p1|1{Y NB. déphasage pour cos dans ]_2p1,0]
ds=. _2p1|1r2p1+1{Y NB. idem pour sin 
tc=.0,(}.dc)%(1+i.y)*(2p1)%T NB. temps retard pour cos
ts=.0,(}.ds)%(1+i.y)*(2p1)%T NB. idem pour sin 
X,({.Y),dc,tc,ds,:ts }}

NB. Série de Fourier limitée de multiples degrés d0 d1 d2 ...dk
NB. d'une fonction f sur un intervalle p,q de période T=q-p avec plot
NB. retourne la liste des fts créées et affichées
NB. x (facultatif) : chaîne contenant des options de plot

NB. Utilisation :   R =. [x]  ( (<'f')   SFMAFF  p,q )   d0 d1 ...dk
SFMAFF =:{{R [ ((R=.(m SFM n)y) AFF) n}}  :  {{R [ x ((R=.(m SFM n)y) AFF) n}} 

NB. ************************************

NB. Outils pour ANALYSEDEFOURIER

NB. Options pour plot (nom)
options=.'titlefont arial 20 bold;title Comparaison de 2 fonctions'
options=.options,';captionfont arial 14 italic bold;pensize 4' 
options=.options,';keyfont arial 14 bold;keypos cto;keystyle lbh;key '

NB. Analyse de Fourier limitée au degré N d'une fonction f de période T=q-p
NB. [p,q] : intervalle de définition

NB. Utilisation :      R =. ((<'f')  ANAF  p,q)  N
ANAF =: {{)c 

z=.*:(<0;0){AB=.((m`:6) SF n)y NB. AB : série de fourier et z : carré de a0
'Ro FIC'=.|:*.j./AB [ T=.-/|.n NB. Ro : Amplitudes des harmoniques
FIC=.0,}._2p1|-FIC NB. FIC : déphasage retard pour série cos dans ]_2p1 , 0]
FIS=.0,}._2p1|FIC+1r2p1 NB. FIS : déphasage retard pour série sin dans ]_2p1 , 0]
UUU=:(*:@(m`:6))"0
Pf =.(100 (UUU INTGLC) n)%T NB. Pf : puissance de f
Pah=. z,}.-:+/*:AB NB. Pah : puissance absolue de chaque harmonique
Prh=. Pah%Pf NB. Prh : idem en pourcentage de puissance de f
Pad=. +/\Pah NB. Pad : puissance absolue expliquée pour chaque degré
Prd=. +/\Prh NB. Prd : idem en pourcentage de puissance de f
((>m),'SF',(":y))=:(AB FPT n)"0 NB. fonction crée : fSFN
poub=.EFFACE <'UUU'
AB;Ro;FIC;FIS;Pf;Pah;Prh;Pad;Prd }}

NB. Utilisation : (( 'f';'fSFN')  GRAPHIQUES  p,q)  ZZ où ZZ =.  ((<'f')  ANAF  p,q)  N
GRAPHIQUES =:{{)c

'AB Ro FIC FIS Pf Pah Prh Pad Prd'=:y
T=.-/|.n [ N=._1+N1=.#Ro                                                                                      
X=.1000 sint n
pd 'reset'
pd 'textfont arial 18 bold italic;textcolor BLACK'
pd 'textc 500 _8x ANALYSE DE FOURIER'
pd 'sub 0 0 1000 _40x;sub 3 1'
pd 'new'



pd 'type line;color blue,red;pensize 3'
pd 'keypos lto;keystyle loh;keyfont arial 13 bold'
pd 'key ',(>0{m),', ',(>1{m)
pd (];((0{m)`:6),:((1{m)`:6)) X
pd 'new'
pd 'title Spectre version cosinus'
pd 'keypos rti;keystyle loh;keyfont arial 13 bold'
pd 'keycolor black,red'
pd 'key Amplitudes, "Déphasages retard"' 
pd 'type stick;color black;pensize 4'
pd (i.N1);Ro
pd 'type stick;color red;pensize 4'
pd (i.N1);FIC
pd 'new'
pd 'title Pourcentage de puissance expliquée'
pd 'type line, marker;color black,red;pensize 3'
pd 'keystyle marker;keymarkers diamond, plus;keypos rmi'
pd 'key "Par degré","Par harmonique"'
pd ((i.N1);(100*Prd))
pd ((i.N1);(100*Prh))
pd 'show'
'' }}

NB. Utilisation :      PF  ((<'f')  Titre   p,q)  N
Titre=: {{)c

L=.#V=.'Fonction: ',(>m),' Puissance: ',(":x),' Fonction crée: ',(>m),'SF',(":y)
<('Série de Fourier ordre: ',(":y),' Intervalle: ',(":n),' Période: ',(":-/|.n)),:V }} 

NB. Notice (nom)
Notice=: 67{.' VARIABLES CREES :'
Notice=:Notice,:67{.' hd : numéro d''harmonique ou de degré de polynôme trigo '
Notice=:Notice, 67{.' a : coefficient de cosinus par harmonique (a=.0{AB)'
Notice=:Notice, 67{.' b : coefficient de sinus par harmonique (b=.1{AB)'
Notice=:Notice, 67{.' Ro : amplitude par harmonique'
Notice=:Notice, 67{.'FIC : déphasage retard pour la version cosinus'
Notice=:Notice, 67{.'FIS : déphasage retard pour la version sinus'
Notice=:Notice, 67{.'Pah : puissance absolue expliquée par harmonique'
Notice=:Notice, 67{.'Prh : puissance relative expliquée par harmonique'
Notice=:Notice, 67{.'Pad : puissance absolue expliquée par degré'
Notice=:Notice, 67{.'Prd : puissance relative expliquée par degré'
Notice=:Notice, 67{.' Pf : puissance de la fonction'
Notice=:Notice, 67{.' Pour obtenir a et b écrire : a=.0{AB et b=.1{AB'
Notice=:<Notice 
    

NB. ************************************

NB. Outil pour CONVOLUTION
NB. NMY (conjonction)
NB. on a g(y) = f(t-y)

NB. Utilisation :       g =: (f   NMY   t)"0 
NMY=:{{u n-y}}

NB. ************************************



NB. Coefficients du polynôme de degré n 
NB. passant par les points (xi ; yi) où i=0 1 2 … n
NB. Utilisation :         C=. x0 x1...xn   POLXY    y0 y1 ... yn

POLXY =: {{y%.x^/i.#x}}"_ _

NB. Fonction en escalier passant par des points (X;Y)
NB. Utilisation : f =: (X;Y)  FESC "0

FESC=:{{)a
pq=.({.,{:)X [ 'X Y'=.m
K=.i.#M=.Y,.~(}:@(],.1&|.))/:~pq,-:@(2&(+/\))X
for_k. K do. 'z0 z1 z2'=.k{M
if. (y>:z0)*.(y<:z1) do. R=.z2 end. end.
R }}

NB. y est-il inclus dans l'intervalle p,q ? R=.0 ou 1
NB. n=0 (]p,q[);     n=1 ([p,q]);      n=2 ([p,q[);     n=3 (]p,q])

NB. Utilisation : R=. ((p,q)  IINT  n)"0 y
IINT=:{{)c

'p q'=.m
select. n
case. 0 do. R=.(y> p)*.(y< q)
case. 1 do. R=.(y>:p)*.(y<:q)
case. 2 do. R=.(y>:p)*.(y< q)
case. 3 do. R=.(y> p)*.(y<:q)
end.                                                            
R }}


